```fsharp
//week 8 intepreter
(* Interpreter for a simple WHILE-language.           MRH 21/10 2013 *)
(* Program skeleton                                                   *)
(* Based on a natural semantics of WHILE                              *)

type AExp =
    (* Arithmetical expressions *)
    | N of int                      (* numbers                  *)
    | V of string                   (* variables                *)
    | Add of AExp * AExp            (* addition                 *)
    | Mul of AExp * AExp            (* multiplication           *)
    | Sub of AExp * AExp            (* subtraction              *)


type BExp =
    (* boolean expressions      *)
    | TT                            (* true                     *)
    | FF                            (* false                    *)
    | Eq of AExp * AExp             (* equality                 *)
    | Lt of AExp * AExp             (* less than                *)
    | Neg of BExp                   (* negation                 *)
    | Con of BExp * BExp            (* conjunction              *)

type Stm =
    (* statements             *)
    | Ass of string * AExp          (* assignment             *)
    | Skip
    | Seq of Stm * Stm              (* sequential composition *)
    | ITE of BExp * Stm * Stm       (* if-then-else           *)
    | While of BExp * Stm           (* while                  *)



type State = Map<string, int>

(* update: string -> int -> State -> State   *)
let update x v s = Map.add x v s

(* A: AExp -> State -> int                   *)
let rec A a s =
    match a with
    | N n -> n
    | V x -> Map.find x s
    | Add (a1, a2) -> A a1 s + A a2 s
    | Mul (a1, a2) -> A a1 s * A a2 s
    | Sub (a1, a2) -> A a1 s - A a2 s

(* B: BExp -> State -> bool                  *)
let rec B b s =
```

```
50      match b with
51      | TT -> true
52      | FF -> false
53      | Eq (a1, a2) -> if A a1 s = A a2 s then true else false
54      | Lt (a1, a2) -> if A a1 s < A a2 s then true else false
55      | Neg (bexp) -> not (B bexp s)
56      | Con (b1, b2) -> B b1 s && B b2 s
57
58  (* I: Stm -> State -> State                          *)
59  let rec I stm s =
60      match stm with
61      | Ass (x, a) -> update x (A a s) s
62      | Skip -> s
63      | Seq (stm1, stm2) -> I stm2 (I stm1 s)
64      | ITE (b, stm1, stm2) -> if B b s then I stm1 s else I stm2 s
65      | While (b, stm1) -> if B b s then I stm (I stm1 s) else s
66
67  let fac =
68      Seq(Ass("y", N 1), While(Neg(Eq(V "x", N 0)), Seq(Ass("y", Mul(V "x", V
          "y")), Ass("x", Sub(V "x", N 1)))))
69
70  (* Define an initial state                          *)
71  let s0 = Map.ofList [ ("x", 4) ]
72
73  (* Interpret the program                            *)
74  let s1 = I fac s0
75
76  (* Inspect the resulting state                      *)
77  Map.find "y" s1
78
```